

Apple Training Series

AppleScript 1-2-3

Aufgaben automatisieren mit dem Mac

Sal Soghoian
Bill Cheeseman



3

Objektreferenzen

Verschachtelte Referenzen

Pfadreferenzen


Aliasreferenzen

UNIX-Referenzen

Datei-URLs

Zusammenfassung der Lektion

Wie geht's weiter?



Lektion 3

Objekte identifizieren

Praktisch alle Skripts haben eine Sache gemeinsam: Sie enthalten *Referenzen* auf skriptfähige Objekte. Ein Skript beispielsweise, das alle Bilder in einem Ordner dreht, enthält Referenzen auf die Bilddateien, die gedreht werden sollen. Ein Skript, das Anmerkungen zu eine Anzahl QuickTime-Filme hinzufügt, enthält Referenzen auf die Filmdateien, die bearbeitet werden sollen. Ein Skript schließlich, das Formate auf Absätze in einer Geschichte anwendet, enthält Referenzen auf die zu bearbeitenden Absätze.

Für AppleScript ist jedes »Ding«, auf das es stößt, ein *Objekt*. Das Referenzieren skriptfähiger Objekte ist eine grundlegende Technik, die alle guten Skriptentwickler kennen und täglich einsetzen. In dieser Lektion erhalten Sie umfassendes Wissen zu der Frage, wie man skriptfähige Objekte mithilfe unterschiedlicher *Objektreferenzen* in AppleScript referenziert.

Da uns der Finder bereits so vertraut ist, werden wir hier den Schwerpunkt setzen. Insbesondere werden wir diejenigen Elemente behandeln, die im Finder und im Programm *System Events* als *Objekte* bezeichnet werden. Aus diesem Grund nennen wir Referenzen manchmal auch *Objektreferenzen*.

Diese Lektion wie auch die vier folgenden untersuchen die Aufgaben des Identifizierens, Suchens und Manipulierens von Objekten mithilfe von AppleScript. Diese Lektion sollten nacheinander gelesen werden, denn sie basieren jeweils auf den Konzepten, die in der vorherigen Lektion dargestellt wurden. Zwar werden in den meisten Beispielen Finder-Objekte wie Dateien und Ordner verwendet, doch gelten viele der beschriebenen Prinzipien gleichermaßen für alle skriptfähigen Objekte und nicht nur solche auf der Festplatte.

Objektreferenzen

Jedes skriptfähige Objekt hat eine Position innerhalb der Objekthierarchie. Ob es sich nun um das dritte Wort im zweiten Absatz eines Dokuments, eine PDF-Datei im Ordner *Dokumente* in Ihrem Benutzerordner auf dem Startvolume oder einen Titel in einer iTunes-Wiedergabeliste handelt: Skriptfähige Objekte existieren stets in anderen skriptfähigen Objekten oder gehören zu diesen.

Das Beschreiben der Position eines Objekts auf einer Festplatte oder in einem Dokument oder einem anderen Objekt ist die Methode, mit der Skripts auf ein Objekt verweisen. Eine Objektreferenz ist also eine Beschreibung der Position eines skriptfähigen Objekts. Zwar gibt es unterschiedliche Möglichkeiten, die Position eines Objekts zu beschreiben, doch bleibt ein wichtiges Grundprinzip bei Skripts immer gleich:

HINWEIS ► Skriptfähige Objekte werden durch ihre Positionen in ihrer Objekthierarchie beschrieben, d.h. wo sie sich in der »Befehlskette« relativ zu anderen Objekten befinden.

Dieses Prinzip gilt gleichermaßen für Filme, Bilder, Absätze, Textcontainer, Kreise, Musiktitel, Dokumentdateien, Programmdateien, Ordner und alles andere.

Skriptentwickler können in Mac OS X fünf Arten von Referenzen verwenden, um Objekte auf einer Festplatte zu referenzieren. Wir bezeichnen diese als *verschachtelte Referenzen*, *Pfadreferenzen*, *Aliasreferenzen*, *POSIX-Pfade* und *Datei-URLs*. Zunächst werden wir in unserer Beschreibung der Referenzen zeigen, wie Sie Dateien und Ordner auf Ihrer Festplatte mit diesen fünf Referenztypen identifizieren.

Verschachtelte Referenzen

In der ersten Lektion dieses Buches haben Sie die Grundlagen von AppleScript kennen gelernt, indem Sie Eigenschaften von Finder-Fenstern manipuliert haben. Eine Eigenschaft, die wir dort untersuchten, hieß `target`. Ihr Wert war eine Referenz auf den Ordner, dessen Inhalt im Finder-Fenster gezeigt wurde.

Hier eine kurze Wiederholung: Geben Sie Skript 3.1 in ein neues Skriptfenster im Skripteditor ein und führen Sie es aus:

```
tell application "Finder" to open home
```

Skript 3.1

Ihr Benutzerordner sollte nun das oberste Finder-Fenster auf Ihrem Schreibtisch sein. Lassen Sie es erst einmal dort liegen. Als Nächstes rufen wir den Wert der Eigenschaft `target` des obersten Fensters mithilfe von Skript 3.2 ab.

```
tell application "Finder" to get the target of the front window
```

Skript 3.2

Im Bereich ERGEBNIS unten im Skriptfenster sehen wir eine Referenz auf den Ordner, dessen Inhalt im obersten Finder-Fenster gezeigt wird. Diese Referenz sollte etwa so aussehen (wobei der Benutzerordner natürlich Ihren Namen und nicht den von Sal tragen wird).:

```
folder "sal" of folder "Users" of startup disk of application "Finder"
```

Wir bezeichnen dies als *verschachtelte* Referenz, weil die Objekte ineinander verschachtelt sind – wie bei den russischen Matrjoschkas.

Eine verschachtelte Referenz verwendet das besitzanzeigende `of`, um anzugeben, dass ein Objekt in einem übergeordneten Objekt enthalten ist oder zu diesem gehört. In diesem Beispiel befindet sich der Ordner *sal* im Ordner *Users*, der sich auf dem Startvolumen befindet, und wird im Programm Finder angezeigt.

Alle skriptfähigen Programme verwenden standardmäßig die verschachtelte Referenz, um ihre skriptfähigen Objekte zu beschreiben. Die Skripts 3.3 bis 3.6 zeigen einige Beispiele.

QuickTime Player:

text frame 12 of track 3 of movie 2 of ↵
application "QuickTime Player"

Skript 3.3

QuarkXPress:

word 5 of paragraph 4 of story 1 of ↵
document "Newsletter" of application "QuarkXPress"

Skript 3.4

iDVD:

button 3 of menu 4 of application "iDVD"

Skript 3.5

Finder:

document file "car.pdf" of folder "Documents" of ↵
folder "sal" of folder "Users" of startup disk of ↵
application "Finder"

Skript 3.6

Wenn der Finder oder irgendein anderes skriptfähiges Programm die angeforderte Position eines skriptfähigen Objekts an ein Skript zurückgibt, tut er dies in der Regel in Form einer verschachtelten Referenz.

HINWEIS ► Wenn Sie diese Skripts im weiteren Verlauf in den Skripteditor eingeben, verwenden Sie in jedem Fall Ihren eigenen Kurznamen anstelle von sal und benutzen Sie statt des Namens von Sals Startvolume (Macintosh HD) den des Ihren. Andernfalls werden die Skripts nicht korrekt funktionieren und womöglich noch nicht einmal übersetzt. Um erfolgreiche Übersetzung und Ausführung zu gewährleisten, müssen Sie jede Anweisung in einen tell-Block für das Programm einschließen. Zudem muss das referenzierte Objekt vorhanden sein (z.B. document "Newsletter").

Pfadreferenzen

Um Informationen zu einem bestimmten Objekt anzufordern und Aktionen damit durchzuführen, übergeben Skripts das Objekt zur Verarbeitung in Form einer Referenz an das übergeordnete Programm. Wenn Sie bestimmte Dateien und Ordner manipulieren, übergeben Skripts Referenzen darauf an den Finder, der unter Mac OS X traditionell zur Behandlung von Festplattenobjekten verwendet wird, oder an das Programm System Events.

Wie Sie gesehen haben, können Referenzen auf Festplattenobjekte manchmal sehr lang und wortreich sein. Eine kürzere Form, die häufig zur Übergabe von Referenzen auf Festplattenobjekte an den Finder verwendet wird, ist die *Pfadreferenz*. Wie eine verschachtelte Finder-Referenz, wird auch eine Pfadreferenz zur Referenzierung eines skriptfähigen Objekts auf der Festplatte verwendet. Allerdings benutzt eine Pfadreferenz einen anderen Ansatz, um auf das skriptfähige Objekt zu verweisen.

Am Anfang einer verschachtelten Referenz steht das skriptfähige Objekt, danach wird die Objekthierarchie nach *oben* durchlaufen.

Pfadreferenzen nutzen den umgekehrten Ansatz: Sie beginnen an der Spitze der Objekthierarchie und arbeiten sich dann nach *unten* zum Zielobjekt vor. Zudem wird im Pfad der Pfadreferenz nicht das besitzanzeigende *of* zur Darstellung einer Kapselung verwendet, sondern es handelt sich um eine durch Doppelpunkte getrennte Liste der Namen der Objekte, die in der »Kette« enthalten sind. Diese ist als Folge von Textzeichen angegeben, die von links nach rechts zu lesen ist.

Exemplarisch wollen wir den Finder um eine Referenz auf den Ordner *Dokumente* in Ihrem Benutzerordner bitten. Geben Sie Skript 3.7 ein und führen Sie es aus.

```
tell application "Finder" to get folder "Documents" of home
```

Skript 3.7

Das im Bereich ERGEBNIS unten im Skriptfenster gezeigte Resultat ist eine verschachtelte Referenz, die wie folgt aussieht:

```
folder "Documents" of folder "sal" of folder "Users" of ↵
  startup disk of application "Finder"
```

Nun konvertieren wir diese verschachtelte Referenz in einen *Pfad*, indem wir `as string` zum vorhandenen Skript hinzufügen. Die Wörter `as string` werden vom Skript als

Anweisung aufgefasst, das Ergebnis der vorherigen Aktion aus seinem Standardformat auf das angegebene Format – in diesem Fall einen Textstring – umzustellen.

```
tell application "Finder" to get folder "Documents" of home as string
```

Skript 3.8

Wenn das Skript ausgeführt wird, sieht das Ergebnis etwa so aus:

```
"Macintosh HD:Users:sal:Documents:"
```

Dies ist ein *Pfad*. Es handelt sich um eine durch Doppelpunkte getrennte Liste mit den Namen der Objekte in der Befehlskette des Zielobjekts, die als von links nach rechts zu lesende Zeichenfolge dargestellt wird. Dabei ist der links stehende Name der des obersten Objekts in der Objekthierarchie – in diesem Fall also die Festplatte, die den Ordner *Dokumente* enthält. Jedes Segment in der Kette ist durch einen Doppelpunkt von den anderen getrennt.

Ein Pfad beginnt mit dem Namen des Laufwerks, das das Zielobjekt enthält, gefolgt vom jeweils nächsten Objekt in hierarchisch absteigender Reihenfolge; am Ende steht dann der Name des Zielobjekts selbst. Jedem Container- oder Ordernamen im Pfad folgt ein Doppelpunkt. Ist das letzte Objekt in der Kette ein Ordner, so wird diesem ebenfalls ein Doppelpunkt nachgestellt, andernfalls wird dieser letzte Doppelpunkt weggelassen.

Beispielsweise würde der Pfad zu einem Ordner *Cars* im Ordner *Dokumente* wie folgt aussehen:

```
"Macintosh HD:Users:sal:Dokumente:Cars:"
```

Der Pfad zur Datei *cars.pdf* im Ordner *Dokumente* hingegen sähe so aus:

```
"Macintosh HD:Users:sal:Dokumente:cars.pdf"
```

TIPP ▶ Anm. d. Ü.: Was Macintosh-Pfade betrifft, kann AppleScript auch mit den lokalisierten Versionen umgehen. Die Ordner *Dokumente* und *Documents* oder *Benutzer* und *Users* beziehen sich jeweils auf das gleiche Verzeichnis. *Solange man über den Finder geht!* Alle anderen Programme *können* lokalisierte Ordernamen unterstützen, *müssen aber nicht*.

UNIX-Pfade wie */Users/sal/Documents* funktionieren mit der lokalisierten Variante der Ordernamen normalerweise gar nicht.

Denken Sie daran:

HINWEIS ► Pfade zu Ordnern enden mit einem Doppelpunkt, Pfade zu Dateien hingegen nicht.

Beachten Sie außerdem, dass der gesamte Pfadstring in Anführungszeichen gesetzt ist. Wie Sie in der ersten Lektion erfahren haben, wird Text – sei es ein Name oder ein Pfad – stets in Anführungszeichen gesetzt, um dem Skripteditor anzuzeigen, dass diese Zeichenfolgen nicht als Befehl zu interpretieren sind.

Um die Erstellung einer Pfadreferenz abzuschließen, müssen Sie nur die Klasse des Zielobjekts vor den Pfadstring setzen.

Skript 3.9 beispielsweise zeigt die Pfadreferenz des Ordners *Dokumente*, wie Sie sie eingeben würden, wenn Sie Befehle an den Finder absetzen.

```
tell application "Finder"
    folder "Macintosh HD:Users:sal:Documents:"
end tell
```

Skript 3.9

Da ein Ordner im Wörterbuch des Finders ein Objekt der Klasse `folder` ist, wird das Wort `folder` der Klasse vorangestellt.

Skript 3.10 zeigt eine Pfadreferenz auf eine Dokumentdatei im Ordner *Dokumente*.

```
tell application "Finder"
    document file "Macintosh HD:Users:sal:Dokumente:cars.pdf"
end tell
```

Skript 3.10

Da eine Datei im Funktionsverzeichnis des Finders ein Objekt der Klasse `document file` ist, werden die Wörter `document file` der Klasse vorangestellt.

Wenn Sie nicht sicher sind, zu welcher Klasse das Zielobjekt gehört, können Sie den Klassennamen des Finders für generische Objekte (`item`) verwenden.

```
tell application "Finder"
    item "Macintosh HD:Users:sal:Dokumente:"
    item "Macintosh HD:Users:sal:Dokumente:cars.pdf"
end tell
```

Skript 3.11

Alle obigen Beispiele sind Pfadreferenzen und werden vom Finder zur Referenzierung von Dateien, Festplatten und Ordnern verwendet.

Um zu veranschaulichen, dass der Finder Pfadreferenzen versteht und entsprechend verarbeitet, geben Sie Skript 3.12 ein und führen Sie es aus (vergessen Sie auch hier nicht, Sals Namen durch den Ihres Benutzerordners und Macintosh HD durch den Namen Ihres Startvolumens zu ersetzen):

```
tell application "Finder" to ¬
    open folder "Macintosh HD:Users:sal:Documents:"
```

Skript 3.12

Da es für den Finder kein Problem ist, eine Pfadreferenz zu verstehen, wird der Ordner *Dokumente* nun auf Ihrem Schreibtisch geöffnet.

Aliasreferenzen

Verschachtelte Referenzen sind leicht nachzuvollziehen, weil ihre Syntax der englischen Sprache stark ähnelt. Allerdings sind verschachtelte Referenzen nur gültig, wenn sie innerhalb ihres Hostprogramms verwendet werden; ein programmübergreifender Einsatz ist hingegen nicht möglich.

So versteht beispielsweise der QuickTime Player den Befehl in Skript 3.13 nicht. Deswegen würde dieses vom Skripteditor nicht übersetzt werden.

```
tell application "QuickTime Player"
    open document file "trailer.mov" of folder "Movies" of ¬
        folder "sal" of folder "Users" of the startup disk
end
```

Skript 3.13

Das liegt daran, dass der QuickTime Player nicht weiß, was ein Ordner oder ein Dokument ist. Die verschachtelte Referenz im Beispiel verwendet Begriffe aus dem Funktionsverzeichnis des Finders, z.B. `document file` und `folder`; diese sind jedoch nicht Bestandteil des Scripting-Wörterbuchs des QuickTime Players.

Pfadreferenzen funktionieren auch nicht in den meisten anderen Programmen, weil sie nur für die Verwendung durch den Finder und einige andere Programme vorgesehen sind.

Damit Referenzen auf Dateien oder Ordner in den meisten anderen Programmen funktionieren, müssen sie ein generisches Format aufweisen, das von allen Anwendungen verarbeitet werden kann: *Aliasreferenzen*.

Eine Referenz im Aliasformat sieht wie eine Pfadreferenz aus, nur wird dem Pfadstring das Wort *alias* anstelle von *document*, *file*, *item* oder *folder* vorangestellt.

Die Skripts 3.14 und 3.15 sind Referenzen auf den Ordner *Dokumente* im Pfad- und Aliasformat.

Pfadreferenz:

```
folder "Macintosh HD:Users:sal:Documents:"
```

Skript 3.14

Aliasreferenz:

```
alias "Macintosh HD:Users:sal:Documents:"
```

Skript 3.15

Der Finder und alle anderen Anwendungen, die Dateien verarbeiten können, verstehen und akzeptieren Referenzen auf Festplattenobjekte im Aliasformat. Schließen Sie alle offenen Finder-Fenster und geben Sie Skript 3.16 ein, bei dem es sich um eine modifizierte Version zuvor aufgeführter Skripts handelt.

```
tell application "Finder" to ¬
    open alias "Macintosh HD:Users:sal:Documents:"
```

Skript 3.16

Aliasreferenz anfordern

Da der Finder Aliasreferenzen verarbeitet, stellt er auf Anforderung eine Referenz auf ein Objekt im Aliasformat bereit. Hierzu fügt er die Coercion *as alias* am Ende der Anforderung ein.

Damit der Finder also etwa eine Aliasreferenz auf den Ordner *Dokumente* in Ihrem Benutzerordner zurückgibt, geben Sie Skript 3.17 ein und führen Sie es aus.

```
tell application "Finder" to get folder "Documents" of home as alias
```

Skript 3.17

Das Ergebnis dieser Anforderung ist eine Aliasreferenz auf den Ordner *Dokumente*:

```
alias "Macintosh HD:Users:sal:Documents:"
```

Zwar können Sie Aliasreferenzen in jedem Programm benutzen, das zur Verarbeitung von Dateien in der Lage ist, doch gibt es einen schwerwiegenden Nachteil in Bezug auf Mac OS X-Versionen vor Leopard.

Damit die Verwendung einer Aliasreferenz in einem Skript für eine Betriebssystemversion vor Leopard zulässig ist, muss sie auf ein *vorhandenes* Objekt verweisen. Ein Skript, das eine Aliasreferenz auf ein nichtexistentes Objekt enthält, kann noch nicht einmal übersetzt werden. Wenn Sie also etwa Mac OS X 10.4 (Tiger) einsetzen, geben Sie die Aliasreferenz aus Skript 3.18 in ein Skriptfenster ein und klicken Sie auf den Button ÜBERSETZEN.

```
alias "Macintosh HD:Users:sal:Documents:zippydodah"
```

Skript 3.18

Der Skripteditor meldet daraufhin einen Fehler, der angibt, dass die Datei nicht gefunden wurde, weil diese Aliasreferenz auf eine Datei verweist, die auf Ihrem Computer nicht vorhanden ist. Diese Einschränkung macht das Schreiben eines Skripts schwierig, wenn dieses ein Objekt enthält, von dem Sie sicher wissen, dass es zwar später – bei Installation und Ausführung des Skripts – verfügbar sein wird, aber noch nicht zum Zeitpunkt der Abfassung des Skripts.

In Leopard können Sie jedoch ein solches Skript mit Aliasreferenzen auf nicht vorhandene Objekte problemlos übersetzen. Natürlich wird das Skript mit einem Fehler beendet, wenn das referenzierte Objekt auch bei der Ausführung des Skripts noch nicht existiert. Doch nun können Sie es zumindest schon einmal schreiben, übersetzen und speichern, auch wenn das eigentliche Zielobjekt noch fehlt.

UNIX-Referenzen

Seit Einführung der UNIX-Unterfütterung für Mac OS X gestaltet sich die Arbeit mit dem Macintosh wesentlich stabiler und hat jede Menge Erweiterungen erfahren. Mac OS X hat die Fähigkeiten von UNIX mit der traditionellen Benutzerfreundlichkeit von Macs kombiniert, und Mac-User haben heute Zugriff auf zusätzliche Funktionen und Formen der Navigation und Manipulation des Dateisystems. So akzeptiert beispielsweise der Befehl GEHE ZUM ORDNER im Finder-Menü GEHE ZU nur Objektreferenzen im UNIX-Format.

Objektreferenzen werden in UNIX als POSIX-Pfade (Portable Operating System for UNIX) angezeigt. Eine solche Referenz im POSIX-Format ähnelt dem Pfadteil von Aliasreferenzen und Pfadreferenzen, doch wird als Trennzeichen ein Schrägstrich statt des Doppelpunkts verwendet. Die folgenden beiden Pfade verweisen also auf dieselbe Datei:

```
"/Users/sal/Documents/cars.pdf"  
"Macintosh HD:Users:sal:Documents:cars.pdf"
```

Wie Sie sehen, referenziert der erste Schrägstrich im POSIX-Pfad das Startvolumen, ohne seinen Namen anzugeben.

Aliasreferenz in POSIX-Pfad umwandeln

Gelegentlich werden Ihre Skripts mit dem UNIX-Aspekt von Mac OS X kommunizieren müssen. Zum Glück stellt AppleScript eine Möglichkeit bereit, Aliasreferenzen in POSIX-Pfade und umgekehrt zu konvertieren. Auf diese Weise kann der Skriptentwickler Skripts sowohl für die traditionelle Mac- als auch für die UNIX-Welt schreiben.

Hier folgt ein Beispiel, in dem ein POSIX-Pfad aus dem Finder abgeleitet wird.

Bringen Sie zunächst den Finder dazu, eine Referenz auf Ihren Ordner *Dokumente* zurückzugeben. Hierzu geben Sie Skript 3.19 ein und führen es aus.

```
tell application "Finder" to get folder "Documents" of home
```

Skript 3.19

Das Ergebnis ist eine verschachtelte Referenz ähnlich der folgenden:

```
folder "Documents" of folder "sal" of folder "Users" of ↵  
startup disk of application "Finder"
```

Nun fügen Sie am Ende des Skripts eine Coercion hinzu, um aus der zurückgegebenen verschachtelten eine Aliasreferenz zu machen (Skript 3.20).

```
tell application "Finder" to get folder "Documents" of home as alias
```

Skript 3.20

Das Ergebnis ist nun eine Referenz auf den Ordner *Dokumente* im Aliasformat:

```
alias "Macintosh HD:Users:sal:Documents:"
```

Da uns das Skript nun eine Aliasreferenz zurückgibt, können Sie aus dem Ergebnis einen POSIX-Pfad extrahieren, indem Sie ihm `POSIX path of` voranstellen (Skript 3.21).

```
tell application "Finder" to -
    get the POSIX path of (folder "Documents" of home as alias)
```

Skript 3.21

Beachten Sie, dass der Bereich des Skripts, der die Ursprungsreferenz generiert und in eine Aliasreferenz umwandelt, nun in Klammern steht. Hierdurch wird dem Skript angegeben, dass die Aktionen in Klammern zuerst durchgeführt werden sollen, woraufhin deren Ergebnis – hier eine Aliasreferenz – die Klammern ersetzt und erst dann die übrigen Skriptaktionen im `tell`-Block ausgeführt werden.

Das Ergebnis des obigen Beispiels sieht wie folgt aus:

```
"/Users/sal/Documents/"
```

Dies ist der POSIX-Pfad des Ordners *Dokumente*.

Um die UNIX-Grundlage von Mac OS X in Aktion zu sehen, geben Sie den Pfad, den Sie gerade vom obigen Skript zurückerhalten haben, ohne Anführungszeichen in das Dialogfeld ein, das angezeigt wird, wenn Sie zum Finder umschalten. Danach schließen Sie alle Fenster und wählen aus dem Menü `GEHE ZU` den Eintrag `GEHE ZUM ORDNER`. Sofort erscheint Ihr Ordner *Dokumente* auf dem Schreibtisch!

Wir haben bereits angemerkt, dass der Name des Startvolumens nicht am Anfang des Pfades angegeben wird. Bei POSIX-Pfaden wird der Name des Startvolumens in Pfaden weggelassen, die auf Objekte auf diesem Laufwerk verweisen. Laufwerksnamen werden hingegen eingebunden, wenn das Ziellaufwerk nicht das Startvolumen ist. Hier beispielsweise sehen wir einen POSIX-Pfad auf eine Datei, die sich auf einem aktivierten Server befindet:

```
"/Volumes/Backup Drive/Clients/Smith Varney/new pamphlet.pdf"
```

POSIX-Pfad in Aliasreferenz umwandeln

Um einen POSIX-Pfad in eine Aliasreferenz zu konvertieren, verwenden Sie zwei Coercions: die erste auf eine Dateireferenz, die nachfolgende dann auf die Aliasreferenz.

Ersetzen Sie beispielsweise das vorhandene Skript im Skriptfenster durch den POSIX-Pfad, der vom folgenden Skript zurückgegeben wird. Stellen Sie dem Pfad die Phrase as POSIX file nach und führen Sie Skript 3.22 aus.

```
"/Users/sal/Documents/" as POSIX file
```

Skript 3.22

Das Ergebnis ist eine Referenz ähnlich der folgenden:

```
file "Macintosh HD:Users:sal:Documents:"
```

Um die Konvertierung in Skript 3.23 abzuschließen, fügen wir als zweite Coercion dem Term as alias am Anweisungsende hinzu.

```
"/Users/sal/Documents/" as POSIX file as alias
```

Skript 3.23

Da Skripts von oben nach unten und von links nach rechts ausgeführt werden, wird die POSIX-Datei-Coercion zuerst ausgeführt, und das Ergebnis wird dann in eine Aliasreferenz umgewandelt:

```
alias "Macintosh HD:Users:sal:Documents:"
```

Wir erläutern die Verwendung von POSIX-Pfaden in Skripten in späteren Lektionen. An dieser Stelle ist es lediglich wichtig zu wissen, was sie sind und wie Sie zwischen Aliasreferenzen und POSIX-Pfaden konvertieren.

HINWEIS ► Das Umwandeln eines Pfadstrings in einen POSIX-Pfad zu einer Datei mittels Coercion funktioniert auch dann, wenn die Datei nicht auf der Festplatte vorhanden ist. Probieren Sie jedoch, ihn in eine Aliasreferenz zu konvertieren, dann generiert das Skript einen Fehler, wenn Sie es ausführen, obwohl die Datei nicht vorhanden ist.

Datei-URLs

Die letzte Form der Objektreferenz ist der *Datei-URL*.

Die robuste Integration des Internets mit Mac OS X trennt nicht mehr scharf zwischen Objekten auf Ihrer Festplatte und solchen im Internet. Wenn Sie beispielsweise iDISK aus dem Menü GEHE ZU im Finder auswählen, wird Ihre iDisk so aktiviert, als ob sie ein weiterer Ordner auf dem Schreibtisch wäre.

Der Finder unterstützt nicht nur Objektreferenzen im verschachtelten Pfad-, Alias- und POSIX-Pfadformat, sondern auch solche im normalen URL-Format, das auch für die Verknüpfung mit Objekten auf Internetseiten verwendet wird.

Geben Sie beispielsweise Skript 3.24 ein und führen Sie es aus.

```
tell application "Finder" to -
  get the url of folder "Documents" of home
```

Skript 3.24

Das Ergebnis ist eine Referenz im HTML-Format ähnlich der folgenden:

```
"file://localhost/Users/sal/Documents/"
```

Um zu sehen, wie eng das Internet und das Mac OS X-Dateisystem miteinander verknüpft sind, geben Sie diesen URL (ohne Anführungszeichen) in Firefox ein – und schon können Sie auf Ihrer Festplatte surfen! (In Safari funktioniert das übrigens nicht, weil Safari in diesem Fall Ihren Ordner *Dokumente* im Finder öffnet.)

Zusammenfassung der Lektion

In dieser Lektion haben wir fünf Arten von Objektreferenzen untersucht, die vom Finder unter Mac OS X verwendet werden: verschachtelte Referenzen, Pfadreferenzen, Aliasreferenzen, POSIX-Pfade und Datei-URLs. Sie verwenden diese Referenzen in den nächsten Lektionen, wenn Sie erfahren, wie Sie bestimmte Objekte suchen.

Bevor wir mit der nächsten Lektion fortfahren, lesen Sie diese Zusammenfassung, um das Wissen über die verschiedenen Formen der Objektreferenzen zu vertiefen.

Objektreferenzen

Skriptfähige Objekte auf Festplatten werden durch ihre Positionen in ihrer Objekthierarchie referenziert, d.h. wo in der »Befehlskette« sie sich befinden. Eine Referenz auf ein Objekt ist eine Beschreibung seiner Position.

Verschachtelte Referenzen

Eine verschachtelte Referenz ist das normale Referenzformat, das von den meisten skriptfähigen Programmen zurückgegeben wird. Die Position eines skriptfähigen Objekts in der Objekthierarchie wird von unten nach oben beschrieben. Hierbei gibt das besitzanzeigende `of` die Kapselung eines Objekts in einem anderen an.

```
tell application "Finder"
  folder "Documents" of folder "sal" of ↵
    folder "Users" of startup disk
end tell
```

Skript 3.25

Pfadreferenzen

Pfadreferenzen werden in Skripts verwendet, um eine Referenz auf ein skriptfähiges Objekt an den Finder zu übergeben. Pfadreferenzen enden auf Textstrings und werden von links nach rechts gelesen. Die Strings umfassen die Namen aller Objekte in der Hierarchie; diese sind von oben nach unten aufgeführt und jeweils durch einen Doppelpunkt voneinander getrennt. Pfadreferenzen beginnen mit dem Namen der Klasse des Zielobjekts – z.B. `folder` oder `document file` – oder dem generischen Begriff `item`.

```
tell application "Finder"
  folder "Macintosh HD:Users:sal:Documents:"
  item "Macintosh HD:Users:sal:Documents:"
  document file "Macintosh HD:Users:sal:Documents:1028.pdf"
  item "Macintosh HD:Users:sal:Documents:1028.pdf"
end tell
```

Skript 3.26

Aliasreferenzen

Aliasreferenzen werden von allen Programmen erkannt, die Dateien verarbeiten können. Sie werden eingesetzt, wenn Objektreferenzen zwischen Programmen übergeben werden.

```
alias "Macintosh HD:Users:sal:Documents:"
alias "Macintosh HD:Users:sal:Documents:1028.pdf"
```

Skript 3.27

POSIX-Pfade

Die UNIX-Aspekte von Mac OS X nehmen Referenzen auf Objekte im POSIX-Pfadformat an. Dieser Referenztyp gibt die Namen der Objekte in der Hierarchie getrennt durch Schrägstriche an:

```
"/Users/sal/Documents/Projects/Cars/new.txt"
```

POSIX-Pfade können wie folgt aus Aliasreferenzen abgeleitet werden:

```
POSIX path of alias "Macintosh HD:Users:sal:Documents:"
--> Rückgabe: "/Users/sal/Documents/"
```

Skript 3.28

Umgekehrt können Aliasreferenzen in POSIX-Pfade umgeformt werden:

```
"/Users/sal/Documents/" as POSIX file as alias
--> Rückgabe: alias "Macintosh HD:Users:sal:Documents:"
```

Skript 3.29

Datei-URLs

Jedes Objekt im Mac OS X-Dateisystem kann im URL-Format (Uniform Resource Locator) beschrieben werden. Dieses Format wird normalerweise zur Referenzierung von Objekten im Internet benutzt.

```
tell application "Finder" to ↵
    get the url of folder "Documents" of home
--> Rückgabe: "file://localhost/Users/sal/Documents/"
```

Skript 3.30

Wie geht's weiter?

In dieser Lektion haben Sie gesehen, wie AppleScript skriptfähige Objekte in der Objekthierarchie identifiziert. Sie haben alles über die unterschiedlichen Objektreferenztypen erfahren, die AppleScript verwendet, wobei der Schwerpunkt erneut auf dem Finder lag. Diese Kenntnisse werden wir nun in der nächsten Lektion nutzen, wenn Sie erfahren, wie Sie bestimmte Objekte mit AppleScript suchen.